TP7_bis

TP7

exercice 1

```
n = 8
A = PolynomialRing(RationalField(), ['X%s'%i for i in
range(1,n+1)], order='lex')
X = A.gens()
A
```

```
Multivariate Polynomial Ring in X1, X2, X3, X4, X5, X6, X7, X8 ove
Rational Field
```

```
ar = [[1,2],[1,5],[1,6],[2,3],[2,4],[2,8],[3,4],[3,8],[4,5],
         [4,7],[5,6],[5,7],[6,7],[7,8]]; ar
```

```
[[1, 2], [1, 5], [1, 6], [2, 3], [2, 4], [2, 8], [3, 4], [3, 8], [
5], [4, 7], [5, 6], [5, 7], [6, 7], [7, 8]]
```

```
L = [X[i]^3 -1 for i in range(n)]
L
```

```
[X1^3 - 1, X2^3 - 1, X3^3 - 1, X4^3 - 1, X5^3 - 1, X6^3 - 1, X7^3
1, X8^3 - 1]
```

```
Lplus = [X[i-1]^2+X[i-1]*X[j-1]+X[j-1]^2 for [i,j] in ar]
Lplus
```

```
[X1^2 + X1*X2 + X2^2, X1^2 + X1*X5 + X5^2, X1^2 + X1*X6 + X6^2, X2
+ X2*X3 + X3^2, X2^2 + X2*X4 + X4^2, X2^2 + X2*X8 + X8^2, X3^2 +
X3*X4 + X4^2, X3^2 + X3*X8 + X8^2, X4^2 + X4*X5 + X5^2, X4^2 + X4*
+ X7^2, X5^2 + X5*X6 + X6^2, X5^2 + X5*X7 + X7^2, X6^2 + X6*X7 +
X7^2, X7^2 + X7*X8 + X8^2]
```

```
L.extend(Lplus)
L
```

```
[X1^3 - 1, X2^3 - 1, X3^3 - 1, X4^3 - 1, X5^3 - 1, X6^3 - 1, X7^3
1, X8^3 - 1, X1^2 + X1*X2 + X2^2, X1^2 + X1*X5 + X5^2, X1^2 + X1*X
+ X6^2, X2^2 + X2*X3 + X3^2, X2^2 + X2*X4 + X4^2, X2^2 + X2*X8 +
X8^2, X3^2 + X3*X4 + X4^2, X3^2 + X3*X8 + X8^2, X4^2 + X4*X5 + X5^
X4^2 + X4*X7 + X7^2, X5^2 + X5*X6 + X6^2, X5^2 + X5*X7 + X7^2, X6^
+ X6*X7 + X7^2, X7^2 + X7*X8 + X8^2]
```

```
I = ideal(L)
I
```

```
Ideal (X1^3 - 1, X2^3 - 1, X3^3 - 1, X4^3 - 1, X5^3 - 1, X6^3 - 1,
X7^3 - 1, X8^3 - 1, X1^2 + X1*X2 + X2^2, X1^2 + X1*X5 + X5^2, X1^2
X1*X6 + X6^2, X2^2 + X2*X3 + X3^2, X2^2 + X2*X4 + X4^2, X2^2 + X2*
+ X8^2, X3^2 + X3*X4 + X4^2, X3^2 + X3*X8 + X8^2, X4^2 + X4*X5 +
```

```
X5^2, X4^2 + X4*X7 + X7^2, X5^2 + X5*X6 + X6^2, X5^2 + X5*X7 + X7^
X6^2 + X6*X7 + X7^2, X7^2 + X7*X8 + X8^2) of Multivariate Polynomi
Ring in X1, X2, X3, X4, X5, X6, X7, X8 over Rational Field
```

```
B = I.groebner_basis()
B
```

```
[X1 - X7, X2 + X7 + X8, X3 - X7, X4 - X8, X5 + X7 + X8, X6 - X8,
X7^2 + X7*X8 + X8^2, X8^3 - 1]
```

les monômes dominants

```
[g.lm() for g in B]
```

```
[X1, X2, X3, X4, X5, X6, X7^2, X8^3]
```

les monômes réduits

```
I.normal_basis()
```

```
[X7*X8^2, X8^2, X7*X8, X8, X7, 1]
```

I est de dimension 0

```
I.dimension()
```

```
0
```

la dimension de l'espace vectoriel quotient A/I

la dilmension est finie donc Z(I) est fini

```
I.vector_space_dimension()
```

```
6
```

I contient un polynôme univarié en chaque indéterminé et ce polynôme est séparable

l'idéal est radiciel par le lemme de Seidenberg

```
[I.elimination_ideal( [X[j] for j in [k for k in range(n) if k <>
i]]   ).gens()   for i in range(n)]
```

```
[[X1^3 - 1], [X2^3 - 1], [X3^3 - 1], [X4^3 - 1], [X5^3 - 1], [X6^3
1], [X7^3 - 1], [X8^3 - 1]]
```

une autre façon de vérifier que I est radiciel

```
I.radical() == I
```

```
True
```

exercice 2

```
restart
```

```
A = PolynomialRing(RationalField(), 3, 'XYZ', order='lex')
(X,Y,Z) = A.gens()
n = A.ngens()
show(A)
```

$\mathbf{Q}[X, Y, Z]$

```
f1 = X^2*Y-Y+Z
f2 = X*Y^2-X+Z
f3 = X+Y+Z-1
I = ideal(f1,f2,f3)
Q = A.quotient(I)
I.dimension()
```

0

```
show(I)
```

$\left(X^2Y - Y + Z, XY^2 - X + Z, X + Y + Z - 1\right)\mathbf{Q}[X, Y, Z]$

```
show(Q)
```

$\mathbf{Q}[X, Y, Z]/\left(X^2Y - Y + Z, XY^2 - X + Z, X + Y + Z - 1\right)\mathbf{Q}[X, Y, Z]$

```
B = I.groebner_basis()
B
```

```
[X + Y + Z - 1, Y^2 - 1/2*Y + 1/2*Z^3 - 7/4*Z^2 - 17/4*Z + 3/2, Y*
- 1/2*Y + 1/2*Z^2 - 3/4*Z + 1/4, Z^4 - 7/2*Z^3 - 15/2*Z^2 + 15/2*Z
3/2]
```

```
LM = [g.lm() for g in B];LM
```

```
[X, Y^2, Y*Z, Z^4]
```

base de Macaulay

```
NB = I.normal_basis()
NB
```

```
[Z^3, Z^2, Z, Y, 1]
```

```
d = I.vector_space_dimension()
d
```

5

h est inversible si et seulement s'il existe k avec kh=1; de plus si k existe il est unique.
on cherche v avec des coefficients indéterminés. x_1,x_2,x_3,x_3,x_5
on travaille donc dans l'extension Q(x_1,x_2,x_3,x_3,x_5) ;

la forme normale de h

```
h = X^5+Y^5+Z^5-5*X^2*Y-5*Y^2*Z-5*Z*X^2-1
```

```
hred = I.reduce(h)
hred
```

    -5*Y + 25/8*Z^4 + 105/16*Z^3 - 15/16*Z^2 - 145/16*Z + 45/16

```
R = PolynomialRing(QQ,5,'a')
a = list(R.gens())
a
show(R)
```

$\mathbf{Q}[a_0, a_1, a_2, a_3, a_4]$

```
K = FractionField(R)
show(K)
```

$\mathrm{Frac}(\mathbf{Q}[a_0, a_1, a_2, a_3, a_4])$

```
KA = A.change_ring(K)
show(KA)
```

$\mathrm{Frac}(\mathbf{Q}[a_0, a_1, a_2, a_3, a_4])[X, Y, Z]$

```
KI = I.change_ring(KA)
show(KI)
```

$\left(X^2Y - Y + Z, XY^2 - X + Z, X + Y + Z - 1\right)\mathrm{Frac}(\mathbf{Q}[a_0, a_1, a_2, a_3, a_4])[X, Y, Z]$

```
k = add(a[i]*KA(NB[i]) for i in range(d))
show(k)
```

$a_3 Y + a_0 Z^3 + a_1 Z^2 + a_2 Z + a_4$

```
KAh = KA(hred)
KAh
```

    (-5)*Y + 25/8*Z^4 + 105/16*Z^3 + (-15/16)*Z^2 + (-145/16)*Z + 45/1

```
w = KAh*k-1
w
```

    (-5*a3)*Y^2 + 25/8*a3*Y*Z^4 + (-5*a0 + 105/16*a3)*Y*Z^3 + (-5*a1 -
    15/16*a3)*Y*Z^2 + (-5*a2 - 145/16*a3)*Y*Z + (45/16*a3 - 5*a4)*Y +
    25/8*a0*Z^7 + (105/16*a0 + 25/8*a1)*Z^6 + (-15/16*a0 + 105/16*a1 +
    25/8*a2)*Z^5 + (-145/16*a0 - 15/16*a1 + 105/16*a2 + 25/8*a4)*Z^4 +
    (45/16*a0 - 145/16*a1 - 15/16*a2 + 105/16*a4)*Z^3 + (45/16*a1 -
    145/16*a2 - 15/16*a4)*Z^2 + (45/16*a2 - 145/16*a4)*Z + 45/16*a4 -

```
wred =KI.reduce(w)
wred
```

    (-5/8*a0 - 5/4*a1 - 5/2*a2 - 55/16*a3 - 5*a4)*Y + (30115/16*a0 +
    3155/8*a1 + 335/4*a2 - 245/8*a3 + 35/2*a4)*Z^3 + (37695/16*a0 +
    4015/8*a1 + 405/4*a2 - 375/8*a3 + 45/2*a4)*Z^2 + (-5665/2*a0 -
    1205/2*a1 - 255/2*a2 + 765/32*a3 - 65/2*a4)*Z + 4735/8*a0 + 505/4*
    + 55/2*a2 - 45/32*a3 + 15/2*a4 - 1

```
LC = wred.coefficients()
```

```
LC
```

```
[-5/8*a0 - 5/4*a1 - 5/2*a2 - 55/16*a3 - 5*a4, 30115/16*a0 +
3155/8*a1 + 335/4*a2 - 245/8*a3 + 35/2*a4, 37695/16*a0 + 4015/8*a1
405/4*a2 - 375/8*a3 + 45/2*a4, -5665/2*a0 - 1205/2*a1 - 255/2*a2 +
765/32*a3 - 65/2*a4, 4735/8*a0 + 505/4*a1 + 55/2*a2 - 45/32*a3 +
15/2*a4 - 1]
```

```
J = Ideal(R(e) for e in LC)
J
```

```
Ideal (-5/8*a0 - 5/4*a1 - 5/2*a2 - 55/16*a3 - 5*a4, 30115/16*a0 +
3155/8*a1 + 335/4*a2 - 245/8*a3 + 35/2*a4, 37695/16*a0 + 4015/8*a1
405/4*a2 - 375/8*a3 + 45/2*a4, -5665/2*a0 - 1205/2*a1 - 255/2*a2 +
765/32*a3 - 65/2*a4, 4735/8*a0 + 505/4*a1 + 55/2*a2 - 45/32*a3 +
15/2*a4 - 1) of Multivariate Polynomial Ring in a0, a1, a2, a3, a4
over Rational Field
```

```
J.dimension()
```

```
0
```

```
J.vector_space_dimension()
```

```
1
```

```
sol = J.variety()
s = sol[0]
s
```

```
{a1: 257691/223892, a2: 1643977/559730, a3: -256/1115, a4:
-1738851/1119460, a0: -202143/559730}
```

```
k = add(s[a[i]]*NB[i] for i in range(d))
k
```

```
-256/1115*Y - 202143/559730*Z^3 + 257691/223892*Z^2 +
1643977/559730*Z - 1738851/1119460
```

on vérifie que k est l'inverse de h dans $\mathbb{Q}[X,Y,Z]/I$

```
I.reduce(h*k)
```

```
1
```

```
I.variety(QQbar)
```

```
[{Z: -2.064177772475913?, Y: 1.532088886237957?, X:
1.532088886237957?}, {Z: 0.3054072893322786?, Y:
0.3472963553338607?, X: 0.3472963553338607?}, {Z:
0.500000000000000?, Y: -0.7807764064044151?, X:
1.280776406404415?}, {Z: 0.500000000000000?, Y:
1.280776406404415?, X: -0.7807764064044151?}, {Z:
4.758770483143633?, Y: -1.879385241571817?, X: -1.879385241571817?
```

malheureusement on trouve les solutions en nombres flottants

on essaie de résoudre le système dans " l'anneau symbolique"

```
x,y,z = var('x,y,z')
```

```
solve([x^2*y-y+z,x*y^2-x+z,x+y+z-1],x,y,z)
```

```
[[x == -1.87938523221, y == -1.87938523221, z == 4.75877046442], [
== 1.53208888889, y == 1.53208888889, z == -2.06417785235], [x ==
0.347296356463, y == 0.347296356463, z == 0.305407287074], [x ==
-4/(sqrt(17) + 1), y == 1/4*sqrt(17) + 1/4, z == (1/2)], [x ==
4/(sqrt(17) - 1), y == -1/4*sqrt(17) + 1/4, z == (1/2)]]
```

pire! on a un mélange de valeurs flottantes et de valeurs exactes
et dans l'aide de solve on trouve cette curieuse "explication"
Whenever possible, answers will be symbolic, but with systems of equations, at times approximations
will be given, due to the underlying algorithm in Maxima !!